Newsgroup: it.comp.software.cad
Titolo: elenco layers in acad

ciao a tutti qualcuno aěsa come si fa a stamparsi l'elenco dei layers in
acad 2000 (ammeeso che sia possibile. avrei bisogno di avere sottocchio e su
cartaceo l'elenco di tutti i layers e possibilmente anche delle loro
caratterisctiche
grazie mille, saluti....

Se i layer non sono molti e ci stanno in una videata oppure, ripetendo
l'operazione piů volte, puoi utilizzare il tasto "Stamp" della tastiera ed
incollare gli appunti in Word.
Altrimenti bisognerebbe scrivere un programma in VisualLisp.
Ciao Claudio

caratterisctiche
La funzione seguente salva su un file l'elenco dei layer con i
seguenti dati:
Nome, Colore,Tipolinea, Stato
Se il colore è uno che ha un nome, viene indicato il nome altimenti il
numero del colore.
Lo stato indica solo se il layer è attivo o disattivato.
Questa è la routine:
--- INIZIO COPIA ---
;
; LAY-FILE.LSP
; COPYRIGHT 2001 CADLandia
;
; Questo Lisp stampa un file testo in formato CSV (con separatore
elenco)
; con l'elenco e gli attributi dei layer nel disegno corrente.
; Il file lista il Nome, il Colore, il Tipolinea e lo Stato dei layer.
; Compariranno i nomi dei colori, se lo hanno, altrimenti il numero.
; Per lo stato si indica SOLO se è attivo o disattivo.
;
; P.S. Il file CSV viene letto da Excel che riconosce i campi.
;
; 31/12/2001 Aniello Annunziata - CADLandia
; 
aniello.a...@cadlandia.com
; 
www.cadlandia.com
(defun C:lay-file ()
;definisce la funzione
  (setq listacolori '("Vuoto" "Rosso" "Giallo" "Verde" "Ciano" "Blu"
"Magenta" "Bianco"))
  ;Lista dei colori. Il primo elemento è inutile
  ; quindi s'è usato il nome "Vuoto"
  (setq fname (getstring "\nInserisci il nome del file: "))
  ;ottiene il nome del file
  (setq fn (open fname "w"))
  ;lo apre in scrittura
  (princ "Nome;Colore;TipoLinea;Stato" fn)
  (princ "\n" fn)
  ;mette l'header del testo
  (setq f 1)
  ;imposta il flag ad 1
  (while
  ;inizia il ciclo while
    (setq t1 (tblnext "Layer" f))
    ;ottiene il nuovo layer
    (setq f nil)
    ;resetta il puntatore
    (setq nome (cdr (assoc 2 t1)))
    ;ottiene il nome del layer
    (setq codcolore (cdr (assoc 62 t1)))
    (if (AND (< codcolore 8) (> codcolore -8))
      (setq colore (nth (abs codcolore) listacolori))
      (setq colore (strcat "Colore " (itoa (abs codcolore))))
    );fine if
    ;ottiene il colore del layer
    (setq tipolinea (cdr (assoc 6 t1)))
    ;ottiene il tipolinea del layer
    (princ nome fn)
    ;stampa il nome
    (princ ";" fn)
    ;stampa uns eparatore elenco
    (princ colore fn)
    ;stampa il colore
    (princ ";" fn)
    ;stampa un separatore elenco
    (princ tipolinea fn)
    ;stampa il tipolinea
    (princ ";" fn)
    ;stampa un separatore elenco
    (if (/= codcolore (abs codcolore))
      (princ "Layer disattivato" fn)
      (princ "Layer attivo" fn)
    );fine if
    ;stampa lo stato del layer
    (princ "\n" fn)
    ;va a capo
  );fine repeat
  (close fn)
  ;chiude il file
  (alert (strcat "Esportati le specifiche dei layer nel file: \n"
         (strcase fname))
  );fine alert
  (princ)
  ;informa l'utente
);fine defun
(princ "\nLAY-FILE.LSP caricato. ")
(princ "Digita LAY-FILE per avviare il comando.")
(princ)
--- FINE COPIA ---
Saluti, Aniello
-- 
Aniello Annunziata
Content Director Manager
www.cadlandia.com
aniello.a...@cadlandia.com
Per rispondermi in privato, togli NOGARBAGE dall'indirizzo